
restart:with(plots):

S. Timoshenko "Analysis of bi-metal thermostats"
Setup: bi-metal stip with thickness h, uniformly heated by DeltaT
consisting of 2 metals with coefficients of expansion alpha_T, Young's moduli E, thicknesses thk,
modified by additional widths

here:
Layer 1 is passive (0 % strain), material PLA
Layer 2 is active with variable % temperature strain, material LAYWOOD meta5

EQUILIBRIUM: 
of forces in layers: P_1 = P_2 (=P)
of moments in layers: P*h/2 = M_1+M_2
internal forces, no external forces
P_1 := P;
P_2 := P;
EQU1 := M_1 + M_2 = P*h/2;

MATERIAL:
M = EI * kappa = EI * 1/r
MAT1 := M_1 = E_1*I_1/r;
MAT2 := M_2 = E_2*I_2/r;

Combining EQU and MAT
EQUMAT:=subs({MAT1,MAT2},EQU1);

KINEMATICS:
on the bearing surfaces the total elongation/strain of both metals must be equal
2 components of strain: 
thermal strain: eps_T = alpha_T_i * Delta_T
mechanical ("elastic") strain:
due to forces: eps_F = P / EA
due to curvature: eps_kappa = kappa * thk_i / 2 = thk_i / (2r)
KIN1 := alpha_T_1*Delta_T + P_1 / (E_1*thk_1*width_1) + thk_1 / 
(2*r) = alpha_T_2*Delta_T - P_2 / (E_2*thk_2*width_2) - thk_2 / 
(2*r);

Combining KIN and (EQU,MAT)
P:=solve(EQUMAT,P);
EQUMATKIN := KIN1;

Second moment of inertia
I_1:=width_1*thk_1^3/12;
I_2:=width_2*thk_2^3/12;

Resulting equation for curvature radius r
r:=simplify(solve(EQUMATKIN,r));

Ratio of Young's moduli n=E_1/E_2
E_1:=n*E_2;
r:=simplify(eval(r)):

Ratio of thicknesses m=thk_1/thk_2 and total height h=thk_1+thk_2
thk_1:=m*h/(1+m);
thk_2:=  h/(1+m);

Ratio of widths q=width_1/width_2
width_1:=q*width_2;

Resulting curvature:
curv:=simplify(1/r);

Timoshenko's original formula (without variable widths)
curvTimo:=6*(alpha_T_2-alpha_T_1)*Delta_T*(1+m)^2/(h*(3*(1+m)^2+
(1+m*n)*(m^2+1/(m*n))));

Data Input 
Bilayer experiment B5 wet-to-dry
for strain evaluation, a negative sign of alpha_T_2 is important

Measured values
Delta_T:=1;
alpha_T_1:=0.0;

width_1:=9.10;
width_2:=20.36;
q:=width_1/width_2;

thk_1:=1.64;
thk_2:=1.36;
m:=thk_1/thk_2;
h:=thk_1+thk_2;

n:=299.705696811487; #from optimization
alpha_T_2:=-0.0509093164628522; #from optimization

curv;

Evaluation of strains

Thermal strain: eps_T = alpha_T_i * Delta_T (constant over thickness)

eps_Temp_1 := alpha_T_1*Delta_T;
eps_Temp_2 := alpha_T_2*Delta_T;

Mechanical/elastic strain 
due to normal forces: eps_F = P / EA with A = thk_i * width (constant over thickness)
due to curvature: eps_kappa = kappa * z_i = z_i / r (linear over thickness, with coordinate z_i 
starting from middle of each layer downwards)

eps_Mech_1F := P_1 / (E_1*thk_1*width_1);
eps_Mech_1M_top := (-thk_1/2)/r;
eps_Mech_1M_mid := 0;
eps_Mech_1M_bottom := (thk_1/2)/r;
eps_Mech_2F := - P_2 / (E_2*thk_2*width_2);
eps_Mech_2M_top := (-thk_2/2)/r;
eps_Mech_2M_mid := 0;
eps_Mech_2M_bottom := (thk_2/2)/r;

Check: total strain on bearing surface (bottom passive and top active) must be equal

eps_Total_1 := eps_Temp_1 + eps_Mech_1F + eps_Mech_1M_bottom;
eps_Total_2 := eps_Temp_2 + eps_Mech_2F + eps_Mech_2M_top;

Plot strains over thickness

Functions over thickness, parameter z starting from bottom upwards
eps_Temp_fct:=piecewise(0<=z and z<thk_2,alpha_T_2*Delta_T,
thk_2<=z and z<=(thk_2+thk_1),alpha_T_1*Delta_T);
eps_Mech_F_fct:=piecewise(0<=z and z<thk_2,-P_2/(E_2*thk_2*
width_2),thk_2<=z and z<=(thk_2+thk_1),P_1/(E_1*thk_1*width_1));
eps_Mech_M_fct:=piecewise(0<=z and z<thk_2,-(-thk_2/2+z)/r,
thk_2<=z and z<=(thk_2+thk_1),-(-thk_2-thk_1/2+z)/r);



> > 

> > 

> > 

(19)(19)

> > 

> > 

> > 

(4)(4)

(11)(11)

(20)(20)

(18)(18)

> > 

> > 

eps_Total_fct:=eps_Temp_fct+eps_Mech_F_fct+eps_Mech_M_fct;

Total strains on top and bottom
eps_Tot_top:=eval(eps_Total_fct,z=h);
eps_Tot_bottom:=eval(eps_Total_fct,z=0);

temperatur strain (red)
plot_T:=plot([eps_Temp_fct,z,z=0..(thk_1+thk_2)],color=red,
legend=['eps_temp']):

normal strain (blue)
plot_F:=plot([eps_Mech_F_fct,z,z=0..(thk_1+thk_2)],color=blue,
linestyle=dash,legend=['eps_normal']):

bending strain (green)
plot_M:=plot([eps_Mech_M_fct,z,z=0..(thk_1+thk_2)],color=green,
linestyle=dash,legend=['eps_bending']):

elastic strain=normal+bending (purple)
plot_elast:=plot([eps_Mech_F_fct+eps_Mech_M_fct,z,z=0..(thk_1+
thk_2)],color=purple,legend=['eps_elastic']):

total strain (black)
plot_Total:=plot([eps_Total_fct,z,z=0..(thk_1+thk_2)],color=
black,linestyle=dashdot,legend=['eps_total']):

display total, elastic and 3 components in one
red: temperature
blue: normal force
red: bending
purple: elastic
black: total
display(plot_elast,plot_F,plot_M,plot_T,plot_Total,labels=

['strain','z']);

>>>high elastic strains of about 5% in the active layer, low total curvature


